


ATARI® 400/800™

AN INVITATION TO PROGRAMMING™ 3:

INTRODUCTION TO SOUND AND GRAPHICS



A Warner Communications Company




Model CX4117

Use with

ATARI® 400™ or ATARI 800™
PERSONAL COMPUTER SYSTEMS

AN INVITATION TO PROGRAMMING™ 3: Introduction to Sound and Graphics



A Warner Communications Company 

Every effort has been made to ensure that this manual accurately documents this product of the ATARI Computer Division. However, because of the ongoing improvement and update of the computer software and hardware, ATARI, INC. cannot guarantee the accuracy of printed material after the date of publication and cannot accept responsibility for errors or omissions.

Reproduction is forbidden without the specific written permission of ATARI, INC., Sunnyvale, CA 94086. No right to reproduce this document, nor the subject matter thereof, is granted unless by written agreement with, or written permission from the Corporation.

CONTENTS

1	INTRODUCTION	1
2	HOW TO USE THIS COURSE	1
	ATARI Components Required	3
	General Instructions for Loading the Cassette Tape	5
	Preparation for the Lesson Practice Sessions	5
3	SUMMARY OF LESSONS: INTRODUCTION TO SOUND	7
4	DESCRIPTION OF FRAMES FOR INTRODUCTION TO SOUND	9
	Lesson One: Sound Registers	9
	Lesson Two: Sound Effects	10
	Lesson Three: Sound Routines	12
	Lesson Four: Efficient Sound Programs	16
	Lesson Five: Musical ABC's	19
	Lesson Six: Identifying Notes	20
	Lesson Seven: Sound Routines With Multiple Registers	23
	Final Quiz: Sound	26
5	SUMMARY OF LESSONS: INTRODUCTION TO GRAPHICS	29
6	DESCRIPTION OF FRAMES FOR INTRODUCTION TO GRAPHICS	31
	Lesson One: Graphics Demonstration	31
	Lesson Two: Changing Color	31
	Lesson Three: Graphics Modes 1 and 2	33
	Lesson Four: Real Graphics	36
	Lesson Five: Tricks With Graphics Modes	39
	Lesson Six: Position and Animation	41
	Lesson Seven: Graphics Characters	45
	Final Quiz: Graphics	47

INTRODUCTION

AN INVITATION TO PROGRAMMING™ 3 is a course designed to introduce concepts in programming sound and graphics. **Introduction to Sound** defines the basics of music theory as well as the characteristics of a computer-generated sound. **Introduction to Graphics** explains the use of color in graphics and introduces the theory of animation.

As this course is the third in a series, effective use of this workbook requires knowledge of ATARI BASIC. Refer to *An Invitation to Programming 1 and 2* or the *ATARI BASIC Reference Manual* for clarification of any BASIC program instructions.

The course has the following components:

- Cassette for Introduction to Sound containing both the computer program and audio material
- Cassette for Introduction to Graphics containing both the computer program and audio material
- This workbook containing both sections of the course to be used as a study guide and reference source

HOW TO USE THIS COURSE

ATARI COMPONENTS REQUIRED

- ATARI® 400™ or ATARI 800™ Personal Computer System with:
8K (minimum) Random Access Memory (RAM)
ATARI BASIC Computing Language Cartridge
- ATARI 410™ Program Recorder
- Introduction to Sound Cassette
- Introduction to Graphics Cassette

GENERAL INSTRUCTIONS FOR LOADING THE CASSETTE TAPE



1. Connect the **ATARI 400** or **ATARI 800 Personal Computer System** to your television set and to a wall outlet as instructed in the Operator's Manual.
2. Make sure that the **ATARI 410 Program Recorder** is properly connected to the computer console and to a wall outlet (see your *ATARI 410 Program Recorder Operator's Manual* for further details, if necessary).

Note: If you have "daisy chained" (connected in series) other ATARI peripherals to your computer console, and do not wish to disconnect them, connect your ATARI 410 Program Recorder to the I/O CONNECTOR of the last unit in the chain.


3. Make sure that at least 8K of RAM is installed in your ATARI Personal Computer System. See the Operator's Manual for Memory Module™ loading instructions.

Note: If an ATARI Disk Drive is connected to the computer console, the Disk Operating System (DOS) and system software use some of the available Random Access Memory (RAM). The amount of RAM used varies with the version of DOS you are using. Take this overhead into account when calculating the amount of RAM required to run a program.

4. Open the cartridge door and insert the ATARI BASIC Cartridge into the cartridge slot. Use the LEFT CARTRIDGE slot on the ATARI 800 Personal Computer System. Close the cartridge door.
5. Turn on your television set. (You may have to adjust the volume on your television set, since the voice from the audio track comes from the television speaker.)
6. Turn on your ATARI Personal Computer System. The POWER switch is on the right side of the computer console.

7. If all equipment is properly connected and turned on, your television screen should display the READY prompt, with the white cursor just below.
8. Press **STOP/EJECT** on your ATARI 410 Program Recorder to open the cassette door.
9. Load Lesson One by holding the cassette tape so that the label (Side 1) is up and the tape leader is facing you. Refer to Figure 1.
10. Slide the cassette into the cassette holder and close the door.
11. If necessary, press **REWIND** and rewind the tape to the beginning. (Set the counter on the Program Recorder to 000.)
12. Type **CLOAD** on the computer keyboard and press . The computer will "beep" once to remind you to press **PLAY** on the Program Recorder.
13. Press **PLAY** and the  key to start the tape. Through the window in the Program Recorder, note that the tape is turning. The beeps and other sounds you hear coming from the television speaker tell you that Lesson One is being loaded into computer memory.

Note: If you have problems loading either of the course programs and you have peripherals in addition to the Program Recorder attached to the computer console, try disconnecting the other peripherals and connecting the Program Recorder directly to the computer console to isolate any problem. If problems persist, consult the *ATARI 410 Program Recorder Operator's Manual*.

14. When the television screen displays the READY prompt, the first program encountered on the cassette, Lesson One, has been loaded into the computer. Make sure that the **PLAY** button on the Program Recorder is still down (and remains down). On the chart provided, make a note of the number on the Program Recorder's counter. Enter this number in the START AUDIO column for Lesson One. If you want to run Lesson One again, simply rewind the tape to this number. As you proceed with the course, similarly keep track of the START AUDIO numbers of the other lessons.
15. Type **RUN** and press  to start Lesson One. From now on, follow the instructions given in the lesson. At the end of Lesson One, make a note of the Program Recorder counter number. Enter this number in the END AUDIO column for Lesson One.
16. **DO NOT** rewind the cassette tape. When you are ready to begin Lesson Two, follow the same procedure as you did to load Lesson One (begin with Step 12). Follow this procedure for all the lessons in the course and for the Final Quiz.

To go to a specific lesson in the course: Completely **REWIND** the tape. Set the counter at 000. Push **FAST FORWARD** on the Program Recorder. Advance the tape until you reach the START LOADING number that you entered on the chart.

To repeat a lesson just completed: **REWIND** the tape to the START AUDIO number on your chart.




Figure 1 Loading a Cassette Tape

PREPARATION FOR THE PRACTICE SESSIONS

Memory in a computer system is divided into two components, ROM and RAM:

ROM. Read-Only Memory contains programs permanently stored in your computer by the manufacturer. This information is essential for computer operation. The ROM programs are called read-only because they can be read but not changed.

RAM. Random Access Memory temporarily stores programs and data in your computer. You can enter information directly into RAM from the keyboard, which allows you to create new programs or bring stored data into your RAM workspace from diskette, cassette, or cartridge.

Turning off the computer console clears all the information stored in RAM. To write new data without turning your console off and on, use the BASIC instruction **NEW** followed by . A **NEW** command clears any information in RAM.

A **CLOAD** command brings a stored program from the cassette into RAM. To clear out this program and prepare for the practice sessions, you **MUST** enter the **NEW** command.

The following statement is repeated at the beginning of each practice session because of its importance.

Note: To clear RAM computer memory, type **NEW** and press before EVERY practice session and between each example program.

SUMMARY OF LESSONS: INTRODUCTION TO SOUND

Lesson One: Sound Registers

Describes various characteristics of a computer-generated sound. You experiment by changing each characteristic, then listen to the resulting sound.

Lesson Two: Sound Effects

Explains the underlying principles in generating routines for simple sound effects such as thunder or explosions. You also learn how to imitate musical instruments.

Lesson Three: Sound Routines

Explains in logical steps how to develop a routine to play computer music.

Lesson Four: Efficient Sound Programs

Introduces DATA statements and other techniques that extend the program developed in Lesson Three.

Lesson Five: Musical ABC's

Discusses pitch and note representation of sounds. Describes staves and clefs.

Lesson Six: Identifying Notes

Various musical sounds are played and shown on the screen. You identify the sounds and their sound value. At the end of the lesson, you translate written music into a computer program.

Lesson Seven: Sound Routines With More Than One Register

Shows how to create music programs and sound effects using several sound registers at once.

Final Quiz: Sound

Quiz covering all of the material taught in this section of the course.

INTRO. TO SOUND	LESSON	CASSETTE RECORDER COUNTER VALUE		
		START LOADING	START AUDIO	END AUDIO
SIDE 1	1	000		
	2			
	3			
	4			
SIDE 2	5	000		
	6			
	7			
	8			
	FINAL TEST			

Figure 2 Program Recorder Counter Numbers for
Lessons in Introduction to Sound

DESCRIPTION OF FRAMES FOR INTRODUCTION TO SOUND

LESSON ONE: SOUND REGISTERS

Frame 1: The Sound Register

Three variables determine the quality of a sound produced by a sound register: pitch, purity, and loudness.

- Pitch represents the height of a sound. On the ATARI computer the pitch values range from 1 to 255. The larger the assigned number, the lower the pitch. (Frame 19 gives the pitch values of musical sounds.)
- Purity (distortion), a variable, allows you to create special sound effects. Purity values range from 0 to 15. As an example, an assigned number 10 creates a pure tone while the number 12 will give you an interesting buzzer sound, especially with a pitch number around 200.
- The last variable controls the loudness (volume) of a tone. Values range from 1 to 15 with an assigned number 1 producing a barely audible sound, and the number 15 producing a fairly loud sound. Number 8 is considered "normal."

Frame 2: Four Sound Registers

The ATARI computer contains four sound registers numbered 0 through 3.

Frame 3: Sound Statement

To produce a sound, enter four variables into a syntax statement: the sound register, pitch value, purity value, and loudness of the sound. Commas must separate the values.

Example:

SOUND 0, 100, 10, 12

where 0 is the sound register,
100 is the pitch value,
10 is the purity value, and
12 is the loudness value.

Frame 4: Sound Shutoff

You must turn the television set to an audible level in order to hear sound. Once turned on, a sound register stays on until you:

1. Assign a loudness value of 0.
2. Program an END instruction. For example,


```
10 SOUND 1, 50, 10, 8
20 FOR X=1 TO 1000: NEXT X
30 END
```
3. Press the **SYSTEM RESET** key on the right side of the keyboard. (**BREAK** will not stop the sound.)
4. Turn off the sound on your television set.

Frame 5: Practice Session for Lesson One

Note: To clear RAM computer memory, type **NEW** and press **RETURN** before EVERY practice session and between each example program.

1. Type this sound instruction into the computer, then press **RETURN**.
SOUND 0, 50, 2, 8
To stop the sound, type **END**.
2. Change the pitch value in the instructions given in step 1. Repeatedly change the value until you have a general idea of the range of values available to you. (Suggestion: refer to Frame 22 for the pitch values of musical sounds.)
3. Similarly experiment with the purity and loudness values.

When you are ready, load Lesson Two on the cassette by typing **CLOAD** and pressing **RETURN** twice. After the **READY** message appears on the screen, type **RUN** and press **RETURN** to start the lesson.

LESSON TWO: SOUND EFFECTS

Frame 6: Summary of Lesson Two

1. Use variables in SOUND instructions.
SOUND 0, X, Y, Z
SOUND 0, PITCH, 10, 8
SOUND 0, 15, 2, LOUD
2. Produce sound effects with the FOR/NEXT loop.
10 FOR PITCH=1 TO 255
20 SOUND 0, PITCH, 10, 8
30 NEXT PITCH
3. Produce a variety of sounds by varying loudness in a FOR/NEXT loop.
10 REM PIANO
20 FOR LOUD=15 TO 0 STEP -1
30 SOUND 0, 15, 0, LOUD
40 NEXT LOUD
4. Popping sounds, explosions, music produced by tapping, and the like are created by rapid variance between loud and soft.

Frame 7: Practice Session for Lesson Two

Note: To clear RAM computer memory, type **NEW** and press **RETURN** before EVERY practice session and between each example program.

Try these sound effects programs.

```
READY
10 REM THUNDER
20 FOR PITCH=1 TO 255
30 SOUND 0,PITCH,8,15
40 NEXT PITCH
■
```

An organ imitation uses two loudness loops in the following program. The second loop rounds out the sound the way an organ does when the player releases the key.

Line 10 in the program sets the pitch to a random number.

```
READY
10 REM STEAM LOCOMOTIVE
20 FOR LOUD=15 TO 0 STEP -1
30 SOUND 0,15,0,LOUD
40 NEXT LOUD
50 GOTO 10
■
```

```
READY
10 REM SIREN GOES UP AND DOWN
20 FOR PITCH=100 TO 10 STEP -1
30 SOUND 0,PITCH,10,8
40 NEXT PITCH
50 FOR PITCH=10 TO 100
60 SOUND 0,PITCH,10,8
70 NEXT PITCH
80 GOTO 20
■
```

```

READY
5 REM PIPE ORGAN
10 PITCH=INT(RND(1)*255)
20 FOR LOUD=0 TO 15
30 SOUND 0,PITCH,10,LOUD
35 NEXT LOUD
40 FOR LOUD=15 TO 0 STEP -1
50 SOUND 0,PITCH,10,LOUD
55 NEXT LOUD
60 GOTO 10

```

```

READY
10 REM PIANO
20 PITCH=INT(RND(0)*155)
30 FOR LOUD=15 TO 0 STEP -1
40 SOUND 0,PITCH,10,LOUD
50 NEXT LOUD
60 GOTO 20

```

Add the following line to play two notes at the same time.

```
45 SOUND 1, PITCH + 32, 10, LOUD
```

When you are ready, load Lesson Three on the cassette by typing **CLOAD** and pressing **RETURN** twice. After the READY message appears on the screen, type **RUN** and press **RETURN** to start the lesson.

LESSON THREE: SOUND ROUTINES

Frame 8: Sound Routine

A program that produces a series of sounds is called a sound routine. Frames 9 through 17 show the evolution of an efficient way to program a sound routine using only one sound register. (Lesson Seven discusses programs that use more than one register.)

Frame 9: Line Numbers

```
READY
100 SOUND 0,81,10,8
110 SOUND 0,64,10,8
120 SOUND 0,53,10,8
130 SOUND 0,64,10,8
140 SOUND 0,60,10,8
150 SOUND 0,47,10,8
160 SOUND 0,60,10,8
170 END
■
```

Frame 10: Adding Delay Statements

Each separate sound must have a delay loop.

```
READY
100 SOUND 0,81,10,8
105 FOR D=1 TO 300:NEXT D
110 SOUND 0,64,10,8
115 FOR D=1 TO 300:NEXT D
120 SOUND 0,53,10,8
125 FOR D=1 TO 300:NEXT D
130 SOUND 0,64,10,8
135 FOR D=1 TO 300:NEXT D
140 SOUND 0,60,10,8
145 FOR D=1 TO 300:NEXT D
150 SOUND 0,47,10,8
155 FOR D=1 TO 300:NEXT D
160 SOUND 0,60,10,8
165 FOR D=1 TO 300:NEXT D
170 END
■
```

Frame 11: Varying Amounts of Delay

```
READY
100 SOUND 0,81,10,8
105 FOR D=1 TO 400:NEXT D
110 SOUND 0,64,10,8
115 FOR D=1 TO 200:NEXT D
120 SOUND 0,53,10,8
125 FOR D=1 TO 400:NEXT D
130 SOUND 0,64,10,8
135 FOR D=1 TO 400:NEXT D
140 SOUND 0,60,10,8
145 FOR D=1 TO 400:NEXT D
150 SOUND 0,47,10,8
155 FOR D=1 TO 200:NEXT D
160 SOUND 0,60,10,8
165 FOR D=1 TO 600:NEXT D
170 END
■
```

Frame 12: Delay Subroutine

```
READY
100 SOUND 0,81,10,8
105 T=400:GOSUB 1000
110 SOUND 0,54,10,8
115 T=200:GOSUB 1000
120 SOUND 0,53,10,8
125 T=400:GOSUB 1000
130 SOUND 0,64,10,8
135 T=400:GOSUB 1000
140 SOUND 0,60,10,8
145 T=400:GOSUB 1000
150 SOUND 0,47,10,8
155 T=200:GOSUB 1000
160 SOUND 0,60,10,8
165 T=600:GOSUB 1000
170 END
1000 FOR D=1 TO T:NEXT D
1010 RETURN
```

Frame 13: PEEK and POKE

For accurate counts in a music routine, you can use the "clock" built into the computer's memory. Telling the computer to locate the clock requires a PEEK statement. Resetting the clock to 0 before each use requires a POKE statement. The following routine replaces the less accurate FOR/NEXT loop.

```
READY
1000 POKE 20,0:REM SET CLOCK TO 0
1010 C=PEEK(20):REM TAKE TIME COUNT AND STORE IT IN C
1020 IF C<T THEN GOTO 1010:REM IF COUNT IS UNDER T THEN LOOK AT CLOCK AGAIN
1030 RETURN:REM STOP COUNTING WHEN C GETS TO T VALUE
```

Frame 14: Practice Session for Lesson Three

Note: To clear RAM computer memory, type **NEW** and press **RETURN** before EVERY practice session and between each example program.

Run the following program to hear the sound produced.

```
READY
100 SOUND 0,81,10,8
110 SOUND 0,64,10,8
120 SOUND 0,53,10,8
130 SOUND 0,64,10,8
140 SOUND 0,60,10,8
150 SOUND 0,47,10,8
160 SOUND 0,60,10,8
170 END
```

Add the following lines and run it again.

```
READY
105 FOR D=1 TO 400:NEXT D
115 FOR D=1 TO 200:NEXT D
125 FOR D=1 TO 400:NEXT D
135 FOR D=1 TO 400:NEXT D
145 FOR D=1 TO 400:NEXT D
155 FOR D=1 TO 200:NEXT D
165 FOR D=1 TO 600:NEXT D
165 FOR D=1 TO 600:NEXT D
```

Replace the separate FOR/NEXT loops with one FOR/NEXT loop subroutine by adding the following lines to your program. Run the program again.

```
READY
105 T=400:GOSUB 1000
115 T=200:GOSUB 1000
125 T=400:GOSUB 1000
135 T=400:GOSUB 1000
145 T=400:GOSUB 1000
155 T=200:GOSUB 1000
165 T=600:GOSUB 1000
1000 FOR D=1 TO T:NEXT D
1010 RETURN
```

Replace the FOR/NEXT subroutine with the following subroutine, and run the program again.

```
1000 POKE 20,0
1010 C=PEEK(20)
1020 IF C<T THEN GOTO 1010
1030 IF C>=T THEN RETURN
```

Note: With T=400, it will take almost seven seconds to play the note! T values should be reduced.

Replace Line 1030 in the above program with the following line and run the program again. Is there any difference? Why?

```
1030 RETURN
```

When you are ready, load Lesson Four on the cassette by typing **CLOAD** and pressing **RETURN** twice. After the READY message appears on the screen, type **RUN** and press **RETURN** to start the lesson.

LESSON FOUR: EFFICIENT SOUND PROGRAMS

Frame 15: Music Program With Timing Subroutine

```
READY
100 PT=81:T=40:GOSUB 1000
110 PT=64:T=20:GOSUB 1000
120 PT=53:T=40:GOSUB 1000
130 PT=64:T=40:GOSUB 1000
140 PT=60:T=40:GOSUB 1000
130 PT=64:T=40:GOSUB 1000
140 PT=60:T=40:GOSUB 1000
150 PT=47:T=20:GOSUB 1000
160 PT=60:T=60:GOSUB 1000
170 END
1000 SOUND 0,PT,10,8
1010 POKE 20,0
1020 C=PEEK(20)
1030 IF C<T THEN GOTO 1020
1040 RETURN
```

Frame 16: Using a READ/DATA Statement

```
READY
100 READ PT,T
110 IF PT=0 AND T=0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 SOUND 0,PT,10,8
1010 POKE 20,0
1020 C=PEEK(20)
1030 IF C<T THEN GOTO 1020
1040 RETURN
2000 DATA 81,40
2010 DATA 64,20
2020 DATA 53,40
2030 DATA 64,50
2040 DATA 60,40
2050 DATA 47,20
2060 DATA 60,60
2070 DATA 0,0
```

Frame 17: More Efficient DATA Lines

```
READY
2000 DATA 81,40,64,20,53,40,64,40,60,40
2010 DATA 47,20,60,60,0,0
■
```

Frame 18: Practice Session for Lesson Four

Note: To clear RAM computer memory, type **NEW** and press **RETURN** before EVERY practice session and between each example program.
EVERY practice session and between each example program.

Our “standard” music program follows. The DATA lines contain the notes from Beethoven’s Fifth Symphony. Run the program as written.

```
READY
100 READ PT,T
110 IF PT=0 AND T=0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 SOUND 0,PT,10,8
1010 POKE 20,0
1020 C=PEEK(20)
1030 IF C<T THEN GOTO 1020
1040 RETURN
2000 DATA 81,30
2010 DATA 81,30
2020 DATA 81,30
2030 DATA 102,90
2040 DATA 0,0
■
```

Something wrong? There are no breaks between the notes! Here is the simplest remedy—add this line to the program:

```
1040 SOUND 0,0,0,0:RETURN
```

Now each note is automatically shut off before the next is played.

Following are the pitch and time values for "Twinkle, Twinkle, Little Star." Enter these into the above program as DATA lines. Make sure that the last DATA line has a pitch value of 0 and a time value of 0 to stop the program.

PT	T
121	30
121	30
60	30
60	30
53	30
53	30
60	60

Here are some songs that you might want to enter into the music program.

```

READY
1999 REM CHRISTMAS THEME
2000 DATA 121,10,121,10,121,20,91,10,9
1,10,91,20,96,10,91,10,81,10,72,10,68,
2000 DATA 121,10,121,10,121,20,91,10,9
1,10,91,20,96,10,91,10,81,10,72,10,68,
10,81,10,72,30
2010 DATA 68,10,60,20,53,10,68,10,72,1
0,91,10,81,20,91,60
2020 DATA 0,0

```

```

READY
1999 REM SAILOR'S WORK SONG
2000 DATA 85,40,85,40,72,40,72,40,108,
40,108,40,144,40,108,20,85,20,72,120
2010 DATA 64,20,85,20,72,120,72,40,85,
40,85,40,72,40,72,40,108,40,108,40,144
,40,108,20,96,20
2020 DATA 85,40,72,80,85,40,96,80,108,
80
2030 DATA 0,0

```

```

READY
1999 REM AMERICA THE BEAUTIFUL
2000 DATA 81,30,81,45,96,15,96,30,81,3
0,81,45,108,15,108,30,96,30,91,30,81,3
0,72,30,64,30,81,90
2010 DATA 81,30,81,45,96,15,96,30,81,3
0,81,45,108,15,108,30,53,30,57,30,53,3
0,47,30,72,30,53,90
2020 DATA 81,30,47,45,47,15,53,30,60,3
0,60,45,64,15,64,30,60,30,53,30,64,30,
72,30,81,30,60,90
2030 DATA 60,30,60,45,72,15,72,30,60,3
0,60,45,81,15,81,30,81,30,72,30,60,30,
81,30,53,30,60,90
2040 DATA 0,0

```

You can use the next program to make the above sounds imitate a piano.

```

READY
10 REM PIANO
100 READ PT,T
110 IF PT=0 AND T=0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 FOR LOUD=15 TO 0 STEP -0.5
1001 SOUND 0,PT,10,LOUD
1002 NEXT LOUD
1010 POKE 20,0
1020 C=PEEK(20)
1010 IF C<2 THEN GOTO 1020
1020 C=PEEK(20)
1030 IF C<T THEN GOTO 1020
1040 RETURN :REM NO SHUTOFF NEEDED

```


When you are ready, load Lesson Five on the cassette. You must turn over the cassette, Introduction to Sound, to Side 2. Refer to General Instructions for Loading the Cassette Tape. Remember to reset the Program Recorder counter to 000.

LESSON FIVE: MUSICAL ABC's

Frame 19: Staves


Notes, represented by symbols, determine the pitch or frequency of musical sounds. The notes are placed on a staff that consists of five lines and four spaces. Each line and each space represent a different pitch.

The two most important staves are the treble staff and the bass staff. A clef sign appears at the beginning of each staff. Three notes—B, C, and D—fall between the two staves.




Frame 20: Additional Information

Music students often use memory devices to help them remember the names of notes. One device: the notes falling in the spaces on the treble clef spell FACE.



Another device: the first letters of the words, Every Good Boy Does Fine, match the names of the notes on the lines.




When you are ready, load Lesson Six on the cassette by typing **CLOAD** and pressing twice. After the READY message appears on the screen, type **RUN** and press to start the lesson.


LESSON SIX: IDENTIFYING NOTES

Frame 21: Ledger Lines

Ledger lines are short lines used for pitches above and below a staff. Here are some ledger lines that can be used to extend the treble staff.




Here are some ledger lines above and below the bass staff.



Frame 22: ATARI Values for Musical Notes

HIGH NOTES	C	29
	B	31
	A# or B \flat	33
	A	35
	G# or A \flat	37
	G	40
	F# or G \flat	42
	F	45
	E	47
	D# or E \flat	50
	D	53
	C# or D \flat	57
	C	60
	B	64
	A# or B \flat	68
	A	72
	G# or A \flat	76
	G	81
	F# or G \flat	85
MIDDLE C	F	91
	E	96
	D# or E \flat	102
	D	108
	C# or D \flat	114
	C	121
	B	128
	A# or B \flat	136
	A	144
	G# or A \flat	153
LOW NOTES	G	162
	F# G \flat	173
	F	182
	D	193
	D# or E \flat	204
	D	217
	C# or D \flat	230
	C	243

Frame 23: Practice Session for Lesson Six

Note: To clear RAM computer memory, type **NEW** and press  before EVERY practice session and between each example program.

Shown below is the music program that we developed in earlier lessons. Use it to play the written music.

```
100 READ PT,T
110 IF PT=0 AND T=0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 SOUND 0,PT,10,8
1010 POKE 20,0
1020 C=PEEK(20)
1030 IF C<T THEN GOTO 1020
1040 SOUND 0,0,0,0:RETURN
2000 DATA .....
```

Time values for this written music are shown below. Use these time values and translate the written music to your computer program so that it can be played.

 = 120  = 60  = 30  = 15

BILLY BOY

Lively



GO TELL AUNT RHODY

Slowly



LOVE SOMEBODY

Lively



When you are ready, load Lesson Seven on the cassette by typing **CLOAD** and pressing **RETURN** twice. After the **READY** message appears on the screen, type **RUN** and press **RETURN** to start the lesson.

LESSON SEVEN: SOUND ROUTINES WITH MULTIPLE REGISTERS

Frame 24: Single Note Music Routine

```
100 READ PT,T
110 IF PT=0 AND T=0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 SOUND 0,PT,10,8
1010 POKE 20,0
1020 C=PEEK(20)
1030 IF C<T THEN GOTO 1020
1040 SOUND 0,0,0,0:RETURN
2000 DATA 81,40
2010 DATA 64,20
2020 DATA 53,40
2030 DATA 64,50
2040 DATA 60,40
2050 DATA 47,20
2060 DATA 60,60
2070 DATA 0,0
2070 DATA 0,0
```

Frame 25: Summary of Lesson Seven

You produce sound effects by turning on or off two or more sound registers. However, for several registers to work in combination to produce useful effects, the registers must blend together or "clash." This usually requires that the sound registers be programmed with pure tones.

Musical notes will blend or clash with other notes depending on the mathematical relationships of the notes' pitch. Notes that are too close together in sound value tend to clash.

A musical chord consists of two or more notes, from more than one register, that blend together.

Frame 26: Practice Session for Lesson Seven

Note: To clear RAM computer memory, type **NEW** and press **RETURN** before EVERY practice session and between each example program.

Try the following sound effects program. This represents two sounds that clash to create a sound effect.

```

READY
5 REM TRAIN HORN
10 FOR HONK=1 TO 2
15 SOUND 0,130,10,8: SOUND 1,175,10,8
20 FOR DELAY=1 TO 750: NEXT DELAY
30 SOUND 0,0,0,0: SOUND 1,0,0,0
40 FOR DELAY=1 TO 50: NEXT DELAY
50 NEXT HONK

```

Here is another effect that works differently.

```

READY
5 REM AIR-RAID SIREN
10 FOR PITCH=100 TO 200
20 SOUND 0,PITCH,10,8
30 SOUND 1,300-PITCH,10,8
40 FOR DELAY=1 TO 5: NEXT DELAY
50 NEXT PITCH
60 GOTO 10
50 NEXT PITCH
60 GOTO 10

```

A program that will play musical chords follows. Enter it and hear the sounds produced. Note that when all four sound registers are turned on, the sound is better if the loudness of each does NOT go over 8.

```

READY
100 READ PT0,PT1,PT2,PT3,T
110 IF PT0=0 AND PT1=0 AND PT2=0 AND PT3=0 AND T=0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 SOUND 0,PT0,10,8: SOUND 1,PT1,10,8
: SOUND 2,PT2,10,8: SOUND 3,PT3,10,8
1010 POKE 20,0
1020 C=PEEK(20)
1030 IF C<T THEN GOTO 1020
1040 SOUND 0,0,0,0: SOUND 1,0,0,0: SOUND
2,0,0,0: SOUND 3,0,0,0: RETURN
2000 DATA 121,96,81,60,20
2010 DATA 162,121,96,81,20
2020 DATA 172,121,96,72,20
2030 DATA 162,121,96,81,40
2040 DATA 198,92,82,64,20
2050 DATA 162,121,96,60,100
2060 DATA 0,0,0,0,0

```

Now enter the following data into the above program to play a familiar tune.

```

READY
2000 DATA 243,193,162,60,20
2010 DATA 243,193,162,60,20
2020 DATA 243,193,162,60,20
2030 DATA 243,193,162,61,20
2040 DATA 182,144,121,72,20
2050 DATA 182,144,121,72,20
2060 DATA 243,193,162,81,40
2070 DATA 0,0,0,47,20
2080 DATA 0,0,0,47,20
2090 DATA 0,128,108,53,20
2100 DATA 0,128,108,53,20
2110 DATA 243,193,162,60,80
3000 DATA 0,0,0,0,0

```

The next program is similar to the last one, except that it sounds more like a piano. You can save time by using the last program for most of the lines and changing only those lines that are different.

```

READY
10 REM PIANO
100 READ PT0,PT1,PT2,PT3,T
110 IF PT0=0 AND PT1=0 AND PT2=0 AND P
T3=0 AND T=0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 FOR LOUD=8 TO 0 STEP -0.5
1002 SOUND 0,PT0,10,LOUD:SOUND 1,PT1,
0,LOUD:SOUND 2,PT2,10,LOUD:SOUND 3,PT3
,10,LOUD
1005 NEXT LOUD
1010 POKE 20,0
1020 C=PEEK(20)
1030 IF C < T THEN GOTO 1020
1040 RETURN :REM SHUT OFF NOT NEEDED
2000 DATA 243,193,162,60,0
2010 DATA 243,193,162,60,0
2020 DATA 243,193,162,60,0
2030 DATA 243,193,162,81,0
2040 DATA 182,144,121,72,0
2050 DATA 182,144,121,72,0
2060 DATA 243,193,162,81,20
2070 DATA 0,0,0,47,0
2080 DATA 0,0,0,47,0
2090 DATA 0,128,108,53,0
2100 DATA 0,128,108,53,0
2110 DATA 243,193,162,60,16
3000 DATA 0,0,0,0,0

```

This program imitates chords on an organ.

```
READY
10 REM ORGAN
100 READ PT0,PT1,PT2,PT3,T
110 IF PT0=0 AND PT1=0 AND PT2=0 AND P
T3=0 AND T=0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 FOR LOUD=0 TO 8
1002 SOUND 0,PT0,10,LOUD:SOUND 1,PT1,1
0,LOUD:SOUND 2,PT2,10,LOUD:SOUND 3,PT3
,10,LOUD
1005 NEXT LOUD
1010 POKE 20,0
1020 C=PEEK(20)
1030 IF C < T THEN GOTO 1020
1036 FOR LOUD=8 TO 0 STEP -0.5
1037 SOUND 0,PT0,10,LOUD:SOUND 1,PT1,1
0,LOUD:SOUND 2,PT2,10,LOUD:SOUND 3,PT3
,10,LOUD
1038 NEXT LOUD
1040 RETURN :REM SHUT OFF NOT NEEDED
2000 DATA 243,193,162,60,20
2010 DATA 243,193,162,60,20
2020 DATA 243,193,162,60,20
2030 DATA 243,193,162,81,20
2040 DATA 182,144,121,72,20
2040 DATA 182,144,121,72,20
2050 DATA 182,144,121,72,20
2060 DATA 243,193,162,81,40
2070 DATA 0,0,0,47,20
2080 DATA 0,0,0,47,20
2090 DATA 0,128,108,53,20
2100 DATA 0,128,108,53,20
2110 DATA 243,193,162,60,80
3000 DATA 0,0,0,0
```

**CONGRATULATIONS! GO ON TO TAKE THE FINAL QUIZ FOR INTRO-
DUCTION TO SOUND. THERE IS NO AUDIO TRACK IN THIS SECTION.**

Frame 27: Use With Final Test

```
READY
100 READ PT0,PT1,PT2,PT3,T
110 IF PT0=0 AND PT1=0 AND PT2=0 AND P
T3=0 AND T=0 THEN END
120 GOSUB 1000
130 GOTO 100
1000 SOUND 0,PT0,10,8:SOUND 1,PT1,10,8
:SOUND 2,PT2,10,8:SOUND 3,PT3,10,8
1010 POKE 20,0
1020 C=PEEK(20)
1030 IF C<T THEN GOTO 1020
1040 SOUND 0,0,0,0:SOUND 1,0,0,0:SOUND
2,0,0,0:SOUND 3,0,0,0:RETURN
2000 DATA 121,96,81,60,20
2010 DATA 162,121,96,81,20
2020 DATA 172,121,96,72,20
2030 DATA 162,121,96,81,40
2040 DATA 108,92,82,64,20
2050 DATA 162,121,96,60,100
2060 DATA 0,0,0,0,0
```

SUMMARY OF LESSONS: INTRODUCTION TO GRAPHICS

Lesson One: Graphics Demonstration

Brief display of simple effects obtained with graphics.

Lesson Two: Changing Color

Instructions in color coding and use of the color registers.

Lesson Three: Graphics Modes 1 and 2

Introduction to graphics modes and in-depth explanation of color registers.

Lesson Four: Real Graphics

Use of PLOT and DRAWTO instructions as well as color in graphic modes.

Lesson Five: Tricks with Graphics Modes

How to use the entire screen in any graphics mode and go from one mode to another without erasing screen memory.

Lesson Six: Position and Animation

How to position text material and give the appearance of movement.

Lesson Seven: Graphics Characters

How to draw pictures with special symbols.

Final Quiz: Graphics

A review of the material covered in this course. The quiz is scored by section.

INTRO. TO GRAPHICS	LESSON	CASSETTE RECORDER COUNTER VALUE		
		START LOADING	START AUDIO	END AUDIO
SIDE 1	1	000		
	2			
	3			
	4			
SIDE 2	5	000		
	6			
	7			
	FINAL TEST			

Figure 3 Program Recorder Counter Numbers for
Lessons in Introduction to Graphics

DESCRIPTION OF FRAMES FOR INTRODUCTION TO GRAPHICS

LESSON ONE: GRAPHICS DEMONSTRATION

When you are ready, load Lesson Two on the cassette by typing **CLOAD** and pressing **ENTER** twice. After the **READY** message appears on the screen, type **RUN** and press **ENTER** to start the lesson.

LESSON TWO: CHANGING COLOR

Frame 1: ATARI BASIC Color Chart

Refer to the following Color Chart for the assigned color number of the approximate color displayed on the screen.

Color Number	Approximate Color on the TV Screen
0	gray
1	gold
2	orange
3	red-orange
4	pink
5	purple or violet
6	red-blue
7	blue
8	blue
9	light blue
10	turquoise
11	green-blue
12	green
13	yellow-green
14	orange-green
15	light orange

Frame 2: Color Registers

The ATARI computer has five color registers numbered 0 through 4. Each color register has a specific application; for example, Color Register 4 changes the color of the frame around the screen and Color Register 2 changes the background color of the screen itself. Two numbers, the color number and its shade value, are associated with each color register.

Frame 3: SETCOLOR Statement

Use the SETCOLOR statement to assign specific colors and associated shades to each color register. The syntax of the SETCOLOR statement is as follows:

SETCOLOR x,y,z where: x represents the color register
y represents the color number
z represents the luminance number

Frame 4: Additional Information on SETCOLOR

Luminance is changed on every EVEN number: 0,2,4,6,8,10,12, and 14.


Note: Luminance Number 10 will cause a blank screen. To correct this situation, reassign a luminance number or press **RETURN**.

Variables can be substituted for numbers in the SETCOLOR statement. An example of the convenience this feature provides follows.

Frame 5: Practice Session for Lesson Two

Note: To clear RAM computer memory, type **NEW** and press **RETURN** before EVERY practice session and between each example program.

Practice using the SETCOLOR statement with either Color Register 2 or 4. Use the direct command mode or the following program to see all the colors and shades available on your ATARI computer.



```
READY
10 REM PROGRAM TO SHOW ALL OF THE COLO
15 AND SHADES
20 REM USE REGISTER 2
30 FOR X=0 TO 15
40 FOR Y=0 TO 14
50 SETCOLOR 2,X,Y
55 FOR I=0 TO 100:NEXT I
60 NEXT Y
70 NEXT X
80 END
```

Experiment with SETCOLOR on Register 1 to determine the shade of the letters on the screen. If the letters disappear while in the direct command mode, press the **SYSTEM RESET** key.

```

READY
10 REM ROCKET
20 SETCOLOR 2,0,2
25 PRINT CHR$(125):REM CLEARS SCREEN
30 FOR COUNT=10 TO 0 STEP -1
40 PRINT COUNT
50 FOR TIME=1 TO 100:NEXT TIME
60 NEXT COUNT
65 SETCOLOR 2,4,14
70 PRINT CHR$(253);"BOOM!"
80 SETCOLOR 2,0,2
90 END

```

When you are ready, load Lesson Three on the cassette by typing **CLOAD** and pressing **RETURN** twice. After the READY message appears on the screen, type **RUN** and press **RETURN** to start the lesson.

LESSON THREE: GRAPHICS MODES 1 AND 2

Frame 6: Graphics Modes

COMPUTER INSTRUCTION	OBJECT ON SCREEN	SIZE (No. of objects on screen)
GRAPHICS 0	regular type	24 lines of 40 characters
GRAPHICS 1	large type (double width)	20 lines of 20 characters
GRAPHICS 2	large type (double height and width)	10 lines of 20 characters
GRAPHICS 3	large graphics squares	20 lines of 40 squares
GRAPHICS 4	smaller graphics points	40 lines of 80 squares
GRAPHICS 5	smaller graphics points (more colors)	40 lines of 80 squares

Frame 7: Color in Graphics Mode 1

Color Registers 2 and 4 activate color choice in Graphics Mode 1. Color Register 2 colors the small text window appearing at the bottom of the screen. Color Register 4 colors the graphics window. Use the SETCOLOR statement to place the color and shade of your choice within the appropriate color register.

SETCOLOR 4, 10, 8

Color Register 4
colors here

LARGE TEXT

SETCOLOR 2, 2, 2

Text window
colored by
Color Register 2

SMALL TEXT

Frame 8: Printing to the Screen in Graphics Mode 1 or 2

Regular PRINT statement instructions appear in the text window portion of the screen. Printing to the graphics portion of the screen requires use of a specific PRINT command. The syntax of this statement is as follows:

PRINT #6; "the print command string enclosed in quotes"

PRINT #6;
"LARGE TEXT"

LARGE TEXT

PRINT
"SMALL TEXT"

SMALL TEXT

Frame 9: Coloring Text in Graphics Window Screen

To have the text characters within the PRINT #6 command appear in assigned colors on the screen, coordinate color registers and character type. As an example:

SETCOLOR 0,9,10 sets the color corresponding to 9, turquoise,
within Color Register 0.

PRINT #6;"TEST" will cause the word TEST to appear in turquoise
on the graphics window screen.

COLOR REGISTER	CHARACTER COLORED	SAMPLE COLOR INSTRUCTION
0	capital letters	SETCOLOR 0,9,10
1	lowercase letters	SETCOLOR 1,15,0
2	inverse capitals	SETCOLOR 2,8,6
3	inverse lowercase	SETCOLOR 3,5,14
4	no character, only background	

Frame 10: Default Colors

If no SETCOLOR commands are given before PRINT #6 instructions are encountered within the program, the character colors will appear in the default color built into the color register.

COLOR REGISTER	CHARACTER COLORED	DEFAULT COLOR
0	capital letters	orange
1	lowercase	light green
2	inverse capital letters	dark blue
3	inverse lowercase	red
4	screen background	dark gray (black)

Frame 11: Practice Session for Lesson Three

Note: To clear RAM computer memory, type **NEW** and press **RETURN** before EVERY practice session and between each example program.

Enter Graphics Mode 1 by typing **GRAPHICS 1** **RETURN**. Type in various PRINT #6; instructions to see what appears on the screen. Do the same with Graphics 2.

Note: Reverse type is produced by first pressing the ATARI symbol key on the lower right side of the keyboard. To stop producing reverse type, press the key a second time. To get back into the regular type mode, type **GRAPHICS 0** **RETURN**

Frame 12: Practice With SETCOLOR in Graphics Modes 1 and 2

Run this program:

```
10 GRAPHICS 2
20 PRINT #6;"ONE two THREE four"
100 END
```

Notice the color of the letters on the graphics window screen. Check the color chart and default values color chart. Now type in the following SETCOLOR commands after the above program has run. Do this while still in graphics mode using the direct command status.

SETCOLOR 0,8,0	observe what happens to: "ONE"
SETCOLOR 1,2,10	observe what happens to: "two"
SETCOLOR 2,15,14	observe what happens to: "THREE"
SETCOLOR 3,0,6	observe what happens to: "four"
SETCOLOR 4,4,2	observe the background screen color

List the above program. Add the following statement lines and run again:

```

30 FOR X=0 TO 3: REM NUMBER OF COLOR REGISTER
40 FOR Y=1 TO 10: REM NUMBER OF FLASHES
50 SETCOLOR X,0,0
60 FOR TIME=1 TO 100:NEXT TIME: REM DELAY
70 SETCOLOR X,0,14
80 FOR TIME=1 TO 100:NEXT TIME: REM DELAY
90 NEXT Y
95 NEXT X

```

Press the **SYSTEM RESET** key. Add the following statement line and run again:

```

15 SETCOLOR 4,4,10

```


Note: Whenever Graphics 0, 1, or 2 is executed, the screen is cleared and default colors are placed in the color registers.

The abbreviation for Graphics is GR, so, Graphics 1 can be typed GR.1.

When you are ready, load Lesson Four on the cassette by typing **CLOAD** and pressing **RETURN** twice. After the READY message appears on the screen, type **RUN** and press **RETURN** to start the lesson.


LESSON FOUR: REAL GRAPHICS

Frame 13: Screen for Graphics Mode 3



Frame 14: PLOT and DRAWTO Commands

To use the PLOT command, visualize a grid pattern on the screen. The PLOT command places a graphics point at the intersection of the x and y coordinates given in the syntax of the command itself. Coordinates can be given from 0 to 39 on the x axis and from 0 to 19 on the y axis.



Frame 15: Sample Program

REMEMBER that the COLOR instruction DOES NOT place a color in any of the registers. The COLOR instruction DOES tell the computer which register will be registers. The COLOR instruction DOES tell the computer which register will be controlling the color of the graphics point.

Determine the missing instruction from the following program:

```
10 GRAPHICS 3
20 SETCOLOR 3,9,4
30 PLOT 15,6
```

Frame 16: COLOR Command and the Color Registers

A COLOR command directly controls the choice of the color register involved in any following PLOT or DRAWTO commands. The following table illustrates this relationship:


COLOR 1	corresponds to Color Register 0
COLOR 2	corresponds to Color Register 1
COLOR 3	corresponds to Color Register 2
COLOR 4	corresponds to Color Register 3

Frame 17: Sample Program Using COLOR Command

```
10 GRAPHICS 3
20 SETCOLOR 0,6,8
30 SETCOLOR 2,0,6
40 COLOR 1
50 PLOT 15,15
```

Remember that color number 0 = gray and 6 = blue.

Frame 18: Practice Session for Lesson Four Using COLOR


Note: To clear RAM computer memory, type **NEW** and press  before EVERY practice session and between each example program.

Run the following program:

```
10 GRAPHICS 3
20 SETCOLOR 0,4,10
30 COLOR 1
40 PLOT 10,10
```

Change the COLOR and SETCOLOR instruction to get different color squares on the screen. Clear memory and enter the following program.

```
10 GRAPHICS 3
20 COLOR 1
30 X=INT(RND(1)*40): Y=INT(RND(1)*20)
40 PLOT X,Y
50 GOTO 20
```

To stop the random graphics display, press the  key. Add these program lines:

```
20 C=INT(RND(1)*4)
25 COLOR C
```

Frame 19: Practice Session for Lesson Four Using PLOT and DRAWTO

Enter and run this program:

```
10 REM FLASHING LIGHTNING PROGRAM
20 GRAPHICS 3
30 COLOR 1
40 PLOT 10,0
42 DRAWTO 20,19
```

Add these additional program lines:

```
45 FOR X=1 TO 10
50 SETCOLOR 0,X,8
60 FOR D=1 TO 100: NEXT D: REM DELAY
70 SETCOLOR 0,9,14
80 FOR D=1 TO 100: NEXT D
100 NEXT X
```

When you are ready, load Lesson Five on the cassette. You must turn over the cassette, Introduction to Graphics, to Side 2. Refer to General Instructions for Loading the Cassette Tape. Remember to reset the Program Recorder counter to 000.

LESSON FIVE: TRICKS WITH GRAPHICS MODES

Frame 20: Appearance of the Screen Using Graphics Modes 3 - 5

	SCREEN	COLOR REGISTERS	MEMORY
GRAPHICS MODE			
3		<p>Available for plot points 0, 1, and 2</p> <p>Background is set by 4</p>	273 bytes
4		<p>Plot points by Plot points by Register 0</p> <p>Background by Register 4</p>	537 bytes 537 bytes
5		<p>Available for plot points 0, 1, and 2</p> <p>Background by Register 4</p>	1017 bytes


Frame 21: Graphics Number + 16

Whenever a text window appears in combination with a main graphics window on the screen, four lines of text are reserved at the bottom of the screen. To use the entire screen for a graphics display, enter the graphics mode and increment its number by 16.

GRAPHICS 3





GRAPHICS 3 + 16



Frame 22: Changing Modes Without Clearing the Screen

GRAPHICS 1





Note: To clear RAM computer memory, type **NEW** and press **RETURN** before EVERY practice session and between each example program.

```

10 REM PRINT TO FULL SCREEN
20 GRAPHICS 2+16
30 FOR X=1 TO 12
40 PRINT #6;"LINE";X
50 NEXT X
60 FOR Y=1 TO 1000: NEXT Y: REM TIME DELAY
70 END

```

```
60 FOR Y=1 TO 1000: NEXT Y: REM TIME DELAY
70 GRAPHICS 1+16+32
```

When you are ready, load Lesson Six on the cassette by typing **CLOAD** and pressing **RETURN** twice. After the READY message appears on the screen, type **RUN** and press **RETURN** to start the lesson.

Frame 24: Sample Program Using POSITION

```
10 GRAPHICS 0
20 SETCOLOR 2,14,10
30 POSITION 12,11
40 PRINT "MIDDLE"
```

Note: POSITION only sets the points where the text will begin on the screen. PRINT, PRINT #6; or DRAWTO commands are necessary before text will actually appear on the screen.

Frame 25: POSITION Instruction

The syntax of the POSITION instruction is:

POSITION x,y where x represents the number across
 y represents the number down

Frame 25: POSITION Instruction

The syntax of the POSITION instruction is:

POSITION x,y where x represents the number across
 y represents the number down

Frame 26: Determining Your Program's Accuracy

PROGRAM 1

```
10 PRINT "TEXT"
20 POSITION 5,5
30 GRAPHICS 0
```

PROGRAM 2
PROGRAM 2

```
10 GRAPHICS 2
20 POSITION 8,4
30 PRINT #6;"text"
```

PROGRAM 3

10 GRAPHICS 1
20 POSITION "TEXT"

Frame 27: Animation Example

```
READY
10 GRAPHICS 1
20 FOR X=1 TO 10
30 POSITION X,5
40 PRINT #6:"WORD"
50 POSITION X,5
60 PRINT #6:" "
70 NEXT X
```

Frame 28: Animation Example

```
READY
10 GRAPHICS 1
20 FOR X=1 TO 10
30 POSITION X,5
40 PRINT H6;"WORD"
45 FOR TIME=1 TO 5:NEXT TIME
50 POSITION X,5
60 PRINT H6;" "
70 NEXT X
■
```

Frame 29: Principles of Animating Text

1. Printed word must be positioned and printed on the screen.
2. Image must be registered in the eye by a slight time delay.
3. Image must be registered in the eye by a slight time delay.
3. Image must be blanked out.
4. Text is repositioned and printed in a slightly different location.

Frame 30: Practice Session for Lesson Six

Note: To clear RAM computer memory, type **NEW** and press **RETURN** before EVERY practice session and between each example program.

Demonstration of POSITION in Graphics 0.

```
READY
10 GRAPHICS 0
20 PRINT "type two numbers--first numb
er is":PRINT "the across value, the se
cond is down:"
30 PRINT:PRINT
40 INPUT X,Y
50 POSITION X,Y
60 PRINT "X POSITION ";X;" ";Y
70 END
■
```

Demonstration of Random Positioning.

```
READY
10 REM PROGRAM TO POSITION AT RANDOM
20 GRAPHICS 2
30 POSITION 5,5
40 PRINT #6;"bubbl9"
50 X=INT(RND(1)*20):Y=INT(RND(1)*10
60 POSITION X,Y:PRINT #6;"0"
70 GOTO 30
■
```

Movement of an arrow across the screen.

```
READY
10 GRAPHICS 2
20 FOR X=1 TO 18
30 POSITION X,5
40 PRINT #6;"->"
50 FOR T=1 TO 100:NEXT T
55 POSITION X,5
60 PRINT #6;"5"
70 NEXT X
■ .....
■
```

Movement of an arrow across the screen without a print command to blank out the image.

Run this program and determine how the image gets blanked out. Explanation is given on the next page.

```
READY
10 GRAPHICS 2
20 FOR X=1 TO 17
30 POSITION X,5
40 PRINT #6;"->"
50 FOR T=1 TO 100:NEXT T
70 NEXT X
■
```

By placing a blank space before the arrow, you create the illusion of movement every time a new position is printed on the screen.

When you are ready, load Lesson Seven on the cassette by typing **CLOAD** and pressing **RETURN** twice. After the READY message appears on the screen, type **RUN** and press **RETURN** to start the lesson.

LESSON SEVEN: GRAPHICS CHARACTERS

Frame 31: POKE Command Instruction




Figure 4 Screen Control Graphics Keyboard Diagram
Figure 4 Screen Control Graphics Keyboard Diagram

To place lowercase and graphics characters on the screen in Graphics Modes 1 or 2, use a POKE command.

1. Enter a specific graphics mode.
2. Use the POKE 756,226 command.
3. PRINT #6;"letters or graphics characters"

Frame 32: Sample Program With Background of Hearts



Frame 33: Sample Program With "Clean Background"

```
READY
10 GRAPHICS 1
20 POKE 756,226
30 SETCOLOR 0,0,0
40 PRINT #6;"GRAPHICS 1H--\`"
```

Frame 34: Practice Session for Lesson Seven

Note: To clear RAM computer memory, type **NEW** and press **RETURN** before EVERY practice session and between each example program.

Enter the following program. The graphics are created as follows: top line = **CTRL** + **M**, bottom line = **CTRL** + **N**
left line = **CTRL** + **V**, right line = **CTRL** + **B**

```
READY
10 GRAPHICS 2
20 POKE 756,226
30 SETCOLOR 0,0,0
40 PRINT #6;"..."
50 PRINT #6;"| lower |"
52 PRINT #6;"| case  |"
60 PRINT #6;"..."
```

Eliminate Line 20 from the above program and run.
Replace Line 20 and eliminate Line 30 and run again.

Demonstration of program animation.

READY

```
10 GRAPHICS 2
20 POKE 756,226
30 SETCOLOR 0,0,0
35 FOR X=0 TO 10
40 POSITION X,5:PRINT #6;"-----"
50 POSITION X,6:PRINT #6;"| lower |"
52 POSITION X,7:PRINT #6;"| case  |"
60 POSITION X,8:PRINT #6;"-----"
65 FOR TIME=1 TO 50:NEXT TIME
70 NEXT X
```

CONGRATULATIONS! GO ON TO THE FINAL QUIZ ON THE CASSETTE.

**LIMITED 90-DAY WARRANTY
ON ATARI® PERSONAL COMPUTER PRODUCTS**

ATARI, INC. ("ATARI") warrants to the original consumer purchaser that this ATARI Personal Computer Product (not including computer programs) shall be free from any defects in material or workmanship for a period of 90 days from the date of purchase. If any such defect is discovered within the warranty period, ATARI's sole obligation will be to repair or replace, at its election, the Computer Product free of charge on receipt of the unit (charges prepaid, if mailed or shipped) with proof of date of purchase satisfactory to ATARI at any authorized ATARI Service Center. For the location of an authorized ATARI Service Center nearest you, call toll-free:

In California (800) 672-1430
Continental U.S. (800) 538-8547

or write to: Atari, Inc.
Customer Service Department
1340 Bordeaux Drive
Sunnyvale, CA 94086

YOU MUST RETURN DEFECTIVE COMPUTER PRODUCTS TO AN AUTHORIZED ATARI SERVICE CENTER FOR IN-WARRANTY REPAIR.

This warranty shall not apply if the Computer Product: (i) has been misused or shows signs of excessive wear, (ii) has been damaged by being used with any products not supplied by ATARI, or (iii) has been damaged by being serviced or modified by anyone other than an authorized ATARI Service Center.

ANY APPLICABLE IMPLIED WARRANTIES, INCLUDING WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE, ARE HEREBY LIMITED TO NINETY DAYS FROM THE DATE OF PURCHASE. CONSEQUENTIAL OR INCIDENTAL DAMAGES RESULTING FROM A BREACH OF ANY APPLICABLE EXPRESS OR IMPLIED WARRANTIES ARE HEREBY EXCLUDED. Some states do not allow limitations on how long an implied warranty lasts or do not allow the exclusion or limitation of incidental or consequential damages, so the above limitations or exclusions may not apply to you.

This warranty gives you specific legal rights and you may also have other rights which vary from state to state.

**DISCLAIMER OF WARRANTY
ON ATARI COMPUTER PROGRAMS**

All ATARI computer programs are distributed on an "as is" basis without warranty of any kind. The entire risk as to the quality and performance of such programs is with the purchaser. Should the programs prove defective following their purchase, the purchaser and not the manufacturer, distributor, or retailer assumes the entire cost of all necessary servicing or repair.

ATARI shall have no liability or responsibility to a purchaser, customer, or any other person or entity with respect to any liability, loss, or damage caused directly or indirectly by computer programs sold by ATARI. This disclaimer includes but is not limited to any interruption of service, loss of business or anticipatory profits or consequential damages resulting from the use or operation of such computer programs.


REPAIR SERVICE

If your ATARI Personal Computer Product requires repair other than under warranty, please contact your local authorized ATARI Service Center for repair information.

IMPORTANT: If you ship your ATARI Personal Computer Product, package it securely and ship it, charges prepaid and insured, by parcel post or United Parcel Service.



PRINTED IN U.S.A.

A Warner Communications Company 

C015727-17 REV. 2